

Paper 232-2010

Pleasing the Client: Creating Custom Reports with SAS® ODS LAYOUT and Proc REPORT

Barbara B. Okerson, HMC, Richmond, VA

ABSTRACT

Through the Output Delivery System (ODS), SAS® software provides a means for creating attractive, concise, and functional reports. Because ODS is an object-oriented (OO) technology, it provides the ability to move beyond template descriptions to creation of complete output layouts. Proc REPORT works with ODS to provide a completely customizable framework for presenting tabular data within the SAS system. The examples in this paper combine text, graphics, and tabular data to create health care management outcomes reports for the PDF destination.

The examples in this paper were developed with version 9.1.3 of SAS executing on a Windows XP platform. The SAS features represented in this paper are not platform-specific and can be adapted by both beginning and advanced SAS users.

INTRODUCTION

Health Management Corporation (HMC) is a WellPoint owned company and is one of the nation's largest, most experienced providers of integrated care and total health solutions. Since 1983, HMC has offered members and clients comprehensive programs and services intended to empower members to take control of their health and see positive results. Through a population-based approach, HMC's disease management programs provide solutions for prevention, chronic condition support, lifestyle management and complex condition care. HMC's mission is to help improve the health and financial outcomes of its clients and members through innovative health solutions that consider every single member at his/her level of care.

To fulfill this mission, HMC provides research, business intelligence, and program evaluation support to measure the effectiveness of the services provided, analyze potential for future products and add to general knowledge of the disease management arena. SAS is the major analytic software used by HMC and, as such, the examples and report layout illustrated in this paper were produced with SAS software in conjunction with this work. SAS software provides all the tools necessary for data validation, analysis, and results reporting, eliminating the need to move output to other software for report formatting.

THE REPORT PROCEDURE

The REPORT procedure was first introduced in version 6 of SAS as a report writing tool to add to the functionality of Proc TABULATE. The procedure combines features of the PRINT, MEANS, and TABULATE procedures with data step manipulations to produce a variety of report types and formats. Proc REPORT can be used in three ways:

- Window environment with a prompting facility that guides report construction.
- Window environment without the prompting facility.
- Non-window (batch) environment (most common today and used in this paper).

Types of reports:

- Detail report containing one row for every observation selected for the report.
- Summary report consolidating data so that each row represents multiple observations.

Both detail and summary reports can contain summary lines as well as detail rows. A summary line summarizes numerical data for a set of detail rows or for all detail rows. These summary lines can be customized. The syntax of Proc REPORT follows. Any break statements or compute blocks follow the last define statement.

```
PROC REPORT data= data-set-name options;  
  COLUMNS variable1 ... variablen;  
  DEFINE variable1;  
  . . .  
  DEFINE variablen;  
RUN;
```

By default, Proc REPORT produces a report similar to one produced by Proc PRINT; the PRINT listing represents the simplest report that can be produced with the report procedure. But the value of Proc REPORT comes from the powerful formatting, summarizing, and analysis capabilities that make it easier to create anything from the simple print listing to a very complex report. With the Output Delivery System, the analysis capabilities are enhanced further with the ability to change every aspect of the look and feel of the report.

OUTPUT DELIVERY SYSTEM

SAS introduced the Output Delivery System (ODS) with the release of version 7; the system became production in version 8. Through ODS, SAS users are provided a multitude of ways to enhance output through formatting functionality not available from individual procedures or from the data step alone. The components of ODS are:

- Procedure output – the text, numbers or graphics results to be formatted
- Table definition – how the output is to be arranged
- Style definition – within that table definition, how colors, fonts, and other appearance factors are to be used
- Destination – where the output is going: display (e.g. HTML, PDF, RTF) or to a dataset

The capabilities of ODS continue to be enhanced. There is support for inline formatting as well as the addition of ODS Layout that provides the flexible page management control needed for today's elaborate executive reports, presentations and dashboards.

ODS LAYOUT

ODS LAYOUT provides the ability to more easily combine text, tables, and graphics in a report within SAS. Absolute layout is supported, with the page divided into regions. ODS LAYOUT in its initial release was limited to a single page; however, page-spanning features are now available for gridded layout. SAS code to use layout consists of ODS statements, global statements, and procedures, but instead of placing the output as streaming output, each piece is placed into the identified region. ODS dimension units can be in centimeters (cm), millimeters (mm), inches (in), printers pt (pt), or pixels (px).

Absolute layout is ideal for reports that use a very precise format and provides flexibility within that context and has been available as experimental since SAS version 9.0. For reports where the size and location of items do not need to be fixed, a relative layout option, gridded layout, is available as experimental beginning in SAS version 9.1. This allows a series of rows and columns of different heights and widths and will also allow the expanding of space if the output does not fit in the "absolute" space. Both the absolute and gridded layout options allow regions to overlap.

ODS LAYOUT is available for the SAS Printer family destinations: PRINTER, PDF, PCL, and PDF. Layout blocks always start with ODS LAYOUT START and end with ODS LAYOUT END. The syntax of ODS LAYOUT is as follows:

```
ODS LAYOUT START options;
ODS REGION options;
SAS Code. . ;
ODS LAYOUT START options;
ODS REGION options;
SAS Code. . ;
ODS LAYOUT END;
```

TASK – PROGRESS REPORT FOR CLINICAL OUTCOMES MEASURES

As part of an effort to improve health care outcomes, clinical measures were being tracked for impact by providers at several geographic locations to catalog current adherence rates as well as identify high and low performers. The request was for a report of progress to date on a measure of particular interest for funding purposes. Patterns for the locations as a whole as well as for individual providers were to be part of the report. This paper traces the evolution from this simple request to a fully formatted report combining text, table, and graphics.

The number of providers and locations are simplified for this example to better illustrate the concepts presented. Health care and health care service data as depicted in this example are for illustration only, were generated for the purpose of this paper, and are not intended to depict any actual results.

Step 1 – Original request.

The original request was to create a table that showed the progress by location and provider for a clinical outcomes measure. The data set used contained variables for the location, provider number, and numerators and denominators for the time periods of the report. Although not used in the original report, change was calculated in the data step before running Proc REPORT. While the original request did not specify what the report should look like, from past experience a spreadsheet look and feel was expected. The style selected for the report was the SAS statistical style. In addition to stock styles such as statistical, styles can be created and modified. Data were grouped around the loc variable (Client Employer Group) which prevented duplication of the row value in repeat observations, as well as providing a variable for summarization grouping. The initial output and Proc REPORT code follows.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance
1	1	189	47%	66%
	2	12	58%	100%
	3	341	49%	89%
	4	150	50%	67%
	5	240	38%	80%
	6	288	49%	85%
	7	246	37%	75%
	8	158	51%	58%
	9	3	33%	100%
	10	2	.00%	100%
	11	5	60%	80%
1		1,634	46%	77%
2	12	541	84%	94%
	13	85	98%	91%
	14	819	78%	77%
	15	732	63%	81%
	16	765	82%	82%
	17	257	7.0%	41%
	18	768	82%	98%
	19	693	86%	88%
	20	571	87%	91%
	21	314	96%	96%
	22	2	.00%	.00%
	23	1	100%	100%
	24	2	100%	100%
	25	2	100%	100%
	26	1	100%	100%
2		5,553	78%	85%
		7,187	70%	83%

```
ods pdf file="r:\bokerson\step1.pdf"
  style=statistical notoc;
Proc REPORT data=one ls=120 ps=40 nowd;
  column loc RN N t1_D t1_N t2_D t2_N t1
  t2;
define loc / group width=9 'Client Employer
Group' center;
define RN / group width=8 'Nurse ID';
define N / display sum format=comma. 'N';
define t1_D / analysis mean noprint;
define t2_D / analysis mean noprint;
define t1_N / analysis mean noprint;
define t2_N / analysis mean noprint;
define t1 / computed format=percent6.2
'Baseline Compliance';
define t2 / computed format=percent6.2
'Current Compliance';
compute t1;
  t1=t1_N.mean/t1_D.mean;
endcomp;
compute t2;
  t2=t2_N.mean/t2_D.mean;
endcomp;
break after loc / summarize ol skip;
rbreak after /summarize dol dul;
run;
ods pdf close;
```

Step 2 – Remove provider detail for low N.

The report provided the information that was requested, but the client did not want to see detail rows for providers with N < 5 patient encounters, but still wanted those included in totals. Additionally the client wanted color changes and cell padding. To accomplish this, those with N > 5 are output from the first report procedure to a new data set and a second Proc REPORT is run. Summary rows are given RN numbers and are formatted to suppress print when higher than 99. Output and new code are below. The COMPUTE statement is used to shade the group summary background rows.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance
1	1	189	47%	66%
	2	12	58%	100%
	3	341	49%	89%
	4	150	50%	67%
	5	240	38%	80%
	6	288	49%	85%
	7	246	37%	75%
	8	158	51%	58%
		1,634	46%	77%
2	12	541	84%	94%
	13	85	98%	91%
	14	819	78%	77%
	15	732	63%	81%
	16	765	82%	82%
	17	257	7.0%	41%
	18	768	82%	98%
	19	693	86%	88%
	20	571	87%	91%
	21	314	96%	96%
		5,553	78%	85%
		7,187	70%	83%

```
data out1;
  set out1;
  if loc=1 and RN=. then RN=101;
  if loc=2 and RN=. then RN=102;
  if loc=. then do;loc=100;RN=201;end;
run;
ods pdf file="r:\bokerson\step2.pdf"
  style=statistical notoc;
Proc REPORT data=out1 ls=120 nowd missing
  style(REPORT)={frame=box font_face=Arial}
  style(REPORT)={outputwidth=80%
  cellpadding=8pt}
  style(HEADER)={background=#8FBC8F}
  style(COLUMN)={foreground=#000000
  background=white}
  style(column)={cellheight=.3in};
column loc RN N t1 t2;
define loc / order order=internal 'Client
Employer Group' center format=maskf.;
define RN /order order=internal 'Nurse ID'
format=maskf.;
define N / sum format=comma. 'N';
define t1 /sum format=percent6.2 'Baseline
Compliance' width=8;
define t2 / sum format=percent6.2 'Current
Compliance' ;
compute t2;
  if RN in (101,102) then call define
  (_row_,"style","style=[background=ltgray
foreground=black]");
  if loc=100 then call define
  (_row_,"style","style=[background=white
foreground=black]");
endcomp; run;
ods pdf close;
```

In the above code, STYLE options are used in the Proc REPORT statement to set several parameters. A table with the most common style elements is included in an appendix at the end of this paper. Within Proc REPORT, style elements can be used in the REPORT statement, DEFINE statement, CALL DEFINE statement, BREAK or RBREAK statement, or in a COMPUTE statement. Using STYLE options in the Proc REPORT statement allows defining style properties for the entire report.

Step 3 – Add progress arrow.

Although the report met all the requested requirements, the client thought it would be much easier to see progress on the indicator if the report included arrows that showed directional movement by each provider. To accomplish this, arrows were created, sized, and written to a bitmap image. Rather than concatenating the arrows to the current cell values, a new column was created to display the arrows. ODS inline formatting with the PREIMAGE style option was used to add the arrows to the report. The POSTIMAGE style option could also be used here as the images do not share space with other elements. The resulting report and code modifications to the second Proc REPORT (above) are illustrated below.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance		
1	1	189	47%	66%	▲	
	2	12	58%	100%	▲	
	3	341	49%	89%	▲	
	4	150	50%	67%	▲	
	5	240	38%	80%	▲	
	6	288	49%	85%	▲	
	7	246	37%	75%	▲	
	8	158	51%	58%	▲	
		1,634	46%	77%	▲	
2	12	541	84%	94%	▲	
	13	85	98%	91%	▼	
	14	819	78%	77%	▼	
	15	732	63%	81%	▲	
	16	765	82%	82%		
	17	257	7.0%	41%	▲	
	18	768	82%	98%	▲	
	19	693	86%	88%	▲	
	20	571	87%	91%	▲	
	21	314	96%	96%		
			5,553	78%	85%	▲
			7,187	70%	83%	▲

While the original data included a change column (difference between baseline and current) two new change variables are necessary because a numeric variable is needed for calculation and a character for display. Change also needs to be a non printed (noprint) column in the first Proc REPORT.

```
column loc RN N t1 t2 change=c2
      change c3;
define change / noprint;
define c2 / noprint;
define c3 / format=$1. ' ' ;
```

New compute statements produce the arrows in the arrow column. The third statement is necessary for cells to remain blank when there is no change.

```
compute c3;
  if c2 > 0 then
    call define (_col_,"style","Style=
[preimage=
'r:\bokerson\up_green.bmp']");
  else if c2 < 0 then
    call define(_col_,"style","Style=
[preimage=
'r:\bokerson\down_red.bmp']");
  else if c2 then
    call define(_col_,"style","style=
[color=black]");
endcomp;
```

Step 4 – Highlight best practices.

While the arrows showed the positive or negative movement over time, this did not show best or highest performers. The client requested an enhancement that would highlight those providers with the highest adherence rate for the measure for each time period. This highlighting needed to occur after those providers with low N's had been removed from the display. As the purpose of highlighting is to make the information of interest stand out from other information, yellow was chosen as the highlight color.

While most data driven highlighting is used to control the background color of table cells based on the value of a cell compared to preset values, in this case, highlighting would be used to compare this value with all other values in the column. Since these numbers would change over time and this code was to be reusable, the ability to compare with the maximum value was required. This was accomplished by running a Proc MEANS, outputting the maximum values for each column to an output dataset and then storing those values in macro variables for use in the Proc REPORT. The code for creating and using these comparison values and the resulting output follows.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance	
1	1	189	47%	66%	📈
	2	12	58%	100%	📈
	3	341	49%	89%	📈
	4	150	50%	67%	📈
	5	240	38%	80%	📈
	6	288	49%	85%	📈
	7	246	37%	75%	📈
	8	158	51%	58%	📈
		1,634	48%	77%	📈
2	12	541	84%	94%	📈
	13	85	98%	91%	📉
	14	819	78%	77%	📉
	15	732	63%	81%	📈
	16	765	82%	82%	
	17	257	7.0%	41%	📈
	18	768	82%	98%	📈
	19	693	86%	88%	📈
	20	571	87%	91%	📈
	21	314	96%	96%	
		5,553	78%	85%	📈
		7,187	70%	83%	📈

Create and format the macro variables and use in compute statements as illustrated below for variable t1.

```
proc means data=out1(where=(RN<99))
  max noprint;
var t1 t2;
output out=t2out max=t1max t2max;
run;
data _null_; set t2out;
call symput('t2max',t2max);
call symput('t1max',t1max);
run;
proc format;
  value t2c &t2max-high=yellow
  other=white;
  value t1c &t1max-high=yellow
  other=white; run;

compute t1;
  if RN in (101,102) then
    call define (_row_,"style",
      style=[background=ltgray
      foreground=black]);
  else if RN ^in (101,102) then
    call define (_col_,"style",
      style=[background=t1c.]);
  if loc=100 then call define
    (_row_,"style",
      style=[background=white
      foreground=black]);
endcomp;
```

Step 5 – Highlight values only for best practices.

Rather than highlighting the entire cell with the highest values, the client wanted the value only to be highlighted. At this time, this is not possible when writing to the SAS ODS PDF destination. But, because RTF (Rich Text Format) has this option in its language, this can be used by SAS. The Rich Text Format (RTF) specification is a method of encoding formatted text and graphics which is read by Microsoft Word and other text editors. SAS allows the ability to embed RTF codes in SAS code. For the most part, these are undocumented.

There are several highlight codes available in RTF, differing only by color. The table below provides this listing. Highlight7 (yellow) was chosen for this report.

Color Hex Code	RTF Name	Color Name
000000	highlight1	Black
0000FF	highlight2	Blue
00FFFF	highlight3	Cyan
00FF00	highlight4	Green
FF00FF	highlight5	Violet
FF0000	highlight6	Red
FFFF00	highlight7	Yellow
FFFFFF	highlight8	White
000080	highlight9	Dark blue
008080	highlight10	Dark cyan
008000	highlight11	Dark green
800080	highlight12	Dark violet
800000	highlight13	Brown
808000	highlight14	Khaki
808080	highlight15	Dark gray
C0C0C0	highlight16	Gray

In addition to figuring out how to write the RTF code, two issues arose concerning the performance arrows. The first was that the RTF destination at this time does not import bitmap images. The second was that the original image arrows, when converted to JPEGs, did not display correctly. New JPEG arrows were created from the Wingdings font, characters 233 and 234.

There are some difficulties with numeric matching in RTF comparisons. While it was originally thought that RTF could only do character matches, a solution was found for a numeric match. Additionally, in order to highlight a field the variable needs to be an order variable. Therefore an additional character variable needed to be created to identify the maximum values. Partial code and output follows.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance		
1	1	189	47%	66%	↑	
	2	12	58%	100%	↑	
	3	341	49%	89%	↑	
	4	150	50%	67%	↑	
	5	240	38%	80%	↑	
	6	288	49%	85%	↑	
	7	246	37%	75%	↑	
	8	158	51%	58%	↑	
		1,634	46%	77%	↑	
2	12	541	84%	94%	↑	
	13	85	98%	91%	↓	
	14	819	78%	77%	↓	
	15	732	63%	81%	↑	
	16	765	82%	82%		
	17	257	7.0%	41%	↑	
	18	768	82%	98%	↑	
	19	693	86%	88%	↑	
	20	571	87%	91%	↑	
	21	314	96%	96%		
			5,553	78%	85%	↑
			7,187	70%	83%	↑

```
proc report data=out1
style(column)={cellheight=.3in};
column loc RN N t1=tla t2=t2a
change=c2 change c3;
define loc/order order=internal
'Client Employer Group' center
format=maskf.;
define RN /order order=internal
'Nurse ID' format=maskf.;
define N/order sum
format=comma.'N';
define tla/order sum
format=percent12.
Style=[protectspecialchars=off]
'Baseline Compliance';
define t2a/order sum
format=percent12.
Style=[protectspecialchars=off]
'Current Compliance';
define change /noprint;
define c2 /noprint;
define c3 / format=5. ' ';

compute tla;
if tla >= &t1max then do;
call define(_col_, "style",
"style=[cellwidth=4 cm pretext=
'\highlight7\insrsid15621608\
charrsid1521608\*' ] ");
;end;
endcomp;

compute t2a;
if t2a >=&t2max then do;
call define(_col_, "style",
"style=[cellwidth= 4 cm pretext=
'\highlight7\insrsid15621608\
charrsid1521608\*' ] ");
end;
```

Step 6 – Include Proc REPORT output in full page final report.

The client wanted the table included in a full-page report with text information and a company banner. For purposes of this paper, the corporate banner has been replaced with a SESUG banner. ODS LAYOUT is used to place the elements correctly on the page. Because a precise layout on a single page is required, absolute layout is used. The results are written to the PDF destination to maintain format and integrity.

The example illustrates use of separate regions for parts of the report as well as the combining of text and table/image within a single region. Some of the other options illustrated in this paper are:

- ODS escapechar – Sets separator character to introduce new syntax and separate options.
- PREIMAGE – Add image before other elements.
- POSTIMAGE – Add image after other elements.
- FONT_FACE – Set font to Arial.
- FONT_SIZE – Set font size to 12 for title, 9 for main text and 7.5 for figure captions.
- FONT_WEIGHT – Set bold on for titles and figure captions.
- JUST – Set justification (center for title – otherwise left).
- STARTPAGE – New procedures should not start new pages. Not currently implemented for Absolute Layout.

Output and the ODS LAYOUT code follows.



Health Care Management Annual Outcomes

Introduction

This report reflects activity for cases under program management during the period 1/1/2008 through 12/31/2008. Participants identified as being in the top future risk percentile and enrolled in the program are assigned to a primary nurse. Once a member is identified and has agreed to participate in ComplexCare, their Nurse Care Manager completes a telephonic assessment to evaluate the member's health status, acuity stratification, and identify care needs. Areas assessed include medication, functional ability, home environment/safety, care access, lifestyle, social/cultural factors, legal/financial factors, and medical history. Nurse/Managers with < 10 patients for the time period were excluded from individual measure calculation but included in the site total.

Client Employer Group	Nurse ID	N	Baseline Compliance	Current Compliance		
1	1	189	47%	66%	↑	
	2	12	58%	100%	↑	
	3	341	49%	89%	↑	
	4	150	50%	67%	↑	
	5	240	38%	80%	↑	
	6	288	49%	85%	↑	
	7	246	37%	75%	↑	
	8	158	51%	58%	↑	
		1,634	46%	77%	↑	
2	12	541	84%	94%	↑	
	13	85	98%	91%	↓	
	14	819	78%	77%	↓	
	15	732	63%	81%	↑	
	16	765	82%	82%		
	17	257	7.0%	41%	↑	
	18	768	82%	98%	↑	
	19	693	86%	88%	↑	
	20	571	87%	91%	↑	
	21	314	96%	96%		
			5,553	78%	85%	↑
			7,187	70%	83%	↑

Please Note

The health quality of program participants improves as adherence to recommended self care practices, testing rates and test results improve. The following sections show results for claims based clinical outcomes for each managed condition. The population includes members in the program with healthcare eligibility throughout the review period for all risk levels. Claims based clinical outcomes typically improve with care management.

```

options nodate nonumber nocenter orientation=portrait spool;
ods pdf file="r:\bokerson\sesug 2009\sesug09.pdf" color=yes style=fancyprinter;
ods pdf startpage=never;
ods layout start width=8 in height=10 in ;
ods region x=20 y=0 width=8 in height=10 in;
ods escapechar="^";
ods pdf text="^S={just=c preimage='R:\bokerson\sesug 2009\sesug2009.bmp^scalewidth'}";
ods pdf text=" ";
ods region x=.5 in y=.8 in width=7 in height=10 in;
ods pdf text="^S={just=c font_weight=bold font_size=12pt font_face=Arial}Health Care
Management Annual Outcomes";
ods pdf text="^S={just=l font_size=9pt font_weight=bold font_face=Arial}Introduction";
ods pdf text="^S={just=l font_size=8pt font_face=Arial}This report reflects activity for
cases under program management during the period 1/1/2008 through 12/31/2008.....";
ods pdf text=" ";
ods pdf text="^S={postimage='R:\Bokerson\sesug 2009\step5.bmp'}";
ods pdf text=" ";
ods pdf text="^S={just=l font_size=9pt font_weight=bold font_face=Arial}Please Note";
ods pdf text="^S={just=l font_size=8pt font_face=Arial}The health quality of program
participants improves as adherence to recommended self care practices, testing rates...";
run;
ods layout end;
ods _all_ close;

```

CONCLUSION

Proc REPORT, ODS, and ODS LAYOUT together provide the ability to easily modify and mix text, graphics, and tables on the same page, arranging them in any position desired on that page. Taking advantage of the large number of formatting options available to enhance the ODS output can make a difference when pleasing a client. And just as important, all this functionality in SAS allows the creation of production quality reports without post-processing. This is especially valuable in the service industry where timely creation of reports, both scheduled and ad hoc, is a major part of securing and retaining business.

REFERENCES

- Chen, Ling Y. "Using v9 ODS Layout to Simplify Generation of Individual Case Summaries", PSUG Proceedings, 2005.
- Haworth, Lauren. "Introduction to ODS," SUGI 29 Proceedings, 2004.
- O'Connor, Daniel, "Next Generation Data _Null_ Report Writing Using ODS OO Features," SUGI 28 Proceedings, 2003.
- O'Connor, Daniel and Scott Huntley, " Breaking New Ground with SAS® 9.2 ODS Layout Enhancements," SAS Global Forum Proceedings, 2009.
- Okerson, Barbara. "Pleasing the Client: Creating Custom Reports with SAS® ODS LAYOUT and Proc REPORT," SESUG Proceedings, 2009.
- Pass, Ray and Sandy McNeill. "PROC REPORT: Doin' It in Style!," SUGI 31 Proceedings, 2006.
- SAS Institute. SAS OnlineDoc 9.1.3. <http://support.sas.com/onlinedoc/913/docMainpage.jsp>
- Schellenberger, Brian T. "ODS LAYOUT: Arranging ODS Output as You See Fit," SUGI 28 Proceedings, 2003.

ACKNOWLEDGEMENTS

I would like to acknowledge the members of the HMC analytic and informatics groups for their suggestions and assistance in the development of this paper. I would also like to thank the SAS Institute technical support team, especially Jody Graham, for assistance and support.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. For more information contact:

Barbara B. Okerson, Ph.D., CPHQ
Senior Health Information Consultant
National Accounts Outcomes
Health Management Corporation (HMC)
8831 Park Central Drive, Suite 100
Richmond, VA 23227
Office: 804-662-5287
Fax: 804-662-5364
Email: bokerson@choosehmc.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX – ODS STYLE ATTRIBUTES

Frequently Used ODS Style Attributes	
Style Attribute	Description
BACKGROUND=	Color of the background of the table or graph
BORDERCOLOR	Color of the border if the border is just one color.
BORDERCOLORDARK	Darker color to use in a border that uses two colors to create 3-D effect.
BORDERCOLORLIGHT	Lighter color to use in a border that uses two colors to create 3-D effect.
BORDERWIDTH	Width of the border of the table.
BOTTOMMARGIN=	Bottom margin for the document.
CELLHEIGHT=	Height of the cell.
CELLPADDING=	Amount of white space on each of the four sides of the text in a cell.
CELLSPACING=	Thickness of the spacing between cells.
CELLWIDTH=	Width of the cell.
CONTRASTCOLOR=	Alternate colors for maps to be applied to the blocks in region areas in block maps.
DROPSHADOW=	Whether to use a drop shadow effect for text in a graph.
ENDCOLOR=	End color for a gradient effect in a graph.
FLYOVER=	Text to show in a tool tip for the cell.
FONT_FACE=	Specify font to use.
FONT_SIZE=	Size of the font to use.
FONT_STYLE=	Style of the font.
FONT_WEIGHT=	Font weight.
FONT_WIDTH=	Font width compared to the width of the usual design.
FONT=	Specify a font definition.
FOREGROUND=	Color of text or data items
IMAGE=	Image to appear in the background. This image can be positioned or tiled.
INDENT=	Set a numeric value to use as the indentation depth.
JUST=	Horizontal positioning for text or images.
LEFTMARGIN=	Left margin for the document.
LINestyle=	Line type to use in a graph (SAS/GRAPH line types).
LINETHICKNESS=	Thickness (width) of a line that is part of a graph.
LINKCOLOR=	Color for links that have not yet been visited.
MARKERSIZE=	Size of the symbol used to represent data values.
MARKERSYMBOL=	Symbol used to represent data values.
NOBREAKSPACE=	How to handle space characters.
OUTPUTHEIGHT=	Height for graphics in the document.
OUTPUTWIDTH=	Width of the table or of the graph or line thickness.
POSTIMAGE=	Image to place after the table or cell.
POSTTEXT=	Text to place after the cell or table.
PREIMAGE=	Image to place before the table or cell.
PRETEXT=	Text to place before the cell or table.
PROTECTSPECIALCHARACTERS=	Determine how <, >, and & are interpreted.
RULES=	Types of rules to use in a table.
STARTCOLOR=	Start color for a gradient effect in a graph.
TOPMARGIN=	Top margin for the document.
TRANSPARENCY=	Level of transparency for a graph.
URL=	Specify a URL to link to.
VJUST=	Vertical justification.